<commit_before>PREFIX rov:    <http://www.w3.org/ns/regorg#>
PREFIX schema: <http://schema.org/>
PREFIX skos:   <http://www.w3.org/2004/02/skos/core#>

DELETE {
  ?organization ?outP ?outO .
  ?inS ?inP ?organization .
}
INSERT {
  ?sampleOrganization ?outP ?outO .
  ?inS ?inP ?sampleOrganization .
}
WHERE {
  {
    SELECT ?name ?postalCode (SAMPLE(?organization) AS ?sampleOrganization)
    WHERE {
      ?organization a schema:Organization ;
        schema:address/schema:postalCode ?postalCode ;
        schema:name|schema:legalName ?name .
      FILTER NOT EXISTS {
        ?organization rov:registration/skos:inScheme <http://linked.opendata.cz/resource/concept-scheme/CZ-ICO> .
      }
    }
    GROUP BY ?name ?postalCode
    HAVING (COUNT(DISTINCT ?organization) > 1)
  }
  ?organization a schema:Organization ;
    schema:address/schema:postalCode ?postalCode ;
    schema:name|schema:legalName ?name .
  FILTER NOT EXISTS {
    ?organization rov:registration/skos:inScheme <http://linked.opendata.cz/resource/concept-scheme/CZ-ICO> .
  }
  FILTER (!sameTerm(?organization, ?sampleOrganization)) 
  ?organization ?outP ?outO .
  OPTIONAL {
    ?inS ?inP ?organization .
  }
}<commit_msg>Organizations have schema:legalName instead of schema:name<commit_after>PREFIX rov:    <http://www.w3.org/ns/regorg#>
PREFIX schema: <http://schema.org/>
PREFIX skos:   <http://www.w3.org/2004/02/skos/core#>

DELETE {
  ?organization ?outP ?outO .
  ?inS ?inP ?organization .
}
INSERT {
  ?sampleOrganization ?outP ?outO .
  ?inS ?inP ?sampleOrganization .
}
WHERE {
  {
    SELECT ?name ?postalCode (SAMPLE(?organization) AS ?sampleOrganization)
    WHERE {
      ?organization a schema:Organization ;
        schema:address/schema:postalCode ?postalCode ;
        schema:legalName ?name .
      FILTER NOT EXISTS {
        ?organization rov:registration/skos:inScheme <http://linked.opendata.cz/resource/concept-scheme/CZ-ICO> .
      }
    }
    GROUP BY ?name ?postalCode
    HAVING (COUNT(DISTINCT ?organization) > 1)
  }
  ?organization a schema:Organization ;
    schema:address/schema:postalCode ?postalCode ;
    schema:legalName ?name .
  FILTER NOT EXISTS {
    ?organization rov:registration/skos:inScheme <http://linked.opendata.cz/resource/concept-scheme/CZ-ICO> .
  }
  FILTER (!sameTerm(?organization, ?sampleOrganization)) 
  ?organization ?outP ?outO .
  OPTIONAL {
    ?inS ?inP ?organization .
  }
}
{'fraction_non_alphanumeric': 0.1307405875051717, 'fraction_numerical': 0.007447248655357882, 'mean_word_length': 3.4944237918215615, 'pattern_counts': {'":': 0, '<': 13, '<?xml version=': 0, '>': 15, 'https://': 0, 'lorem ipsum': 0, 'www.': 4, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5995996', 'n_tokens_mistral': 896, 'n_tokens_neox': 818, 'n_words': 155}